
Change 1
New function f_IMS_INVITE_CalleeUriRX
	  function f_IMS_INVITE_CalleeUriRX(template SipUrl p_OtherCalleeUri := omit,

                                    INVITE_A_2_1_Context_Type p_Context := A_2_1_A4) return template (present) SipUrl

  {

    var template (present) SipUrl v_ExpectedCalleeUri;

    if (ispresent(p_OtherCalleeUri)) {

      v_ExpectedCalleeUri := p_OtherCalleeUri;    /* @sic R5s150204 sic@ */

    } else {

      select (p_Context) {

        case (A_2_1_A4) {                // A4: INVITE creating a dialog

          v_ExpectedCalleeUri := f_SIP_BuildSipUri_RX(px_IMS_CalleeUri);

        }

        case (A_2_1_A5) {                // A5: re-INVITE

          v_ExpectedCalleeUri := f_SIP_BuildSipUri_RX(px_IMS_CalleeContactUri, *);  /* @sic R5s140020 change 8.1 sic@ */

        }

        case (A_2_1_A6, A_2_1_A7, A_2_1_A19) {      // A6, A7, 19: emergency call /* @sic R5-174545, Change 1 R5s180463 sic@ */

          v_ExpectedCalleeUri := cr_SipUrl_Common(c_urnScheme, cr_SipUrl_SOS);

        }

        case else {

          FatalError(__FILE__, __LINE__, "");

        }

      }

    }

    return v_ExpectedCalleeUri;

  }


New function f_IMS_INVITE_ReceiveRequest_Optional
	  function f_IMS_INVITE_ReceiveRequest_Optional(template (present) SipUrl p_ExpectedCalleeUri,

                                                template (present) MessageHeader p_MessageHeader_Invite,

                                                template (present) MessageBody p_MessageBody,

                                                template (omit) float p_WaitDuration := omit,

                                                boolean p_IsReInvite := false,

                                                boolean p_IsEmergencyCallWithoutRegistration := false)

    runs on IMS_PTC return template (omit) IMS_DATA_REQ

  {

    var IMS_DATA_REQ v_IMS_DATA_REQ_ByRef;

    var INVITE_Request v_InviteRequest;

    var SipUrl v_ContactUrl;

    var boolean v_TimerControl := ispresent(p_WaitDuration);

    timer t_Timer;

    if (v_TimerControl) {

      t_Timer.start(valueof(p_WaitDuration));

    }

    alt {

      [] a_IMS_ReceiveRequest(car_IMS_Invite_Request(cr_INVITE_Request(p_ExpectedCalleeUri, p_MessageHeader_Invite, p_MessageBody)), v_IMS_DATA_REQ_ByRef)

        {

          t_Timer.stop;

          v_InviteRequest := v_IMS_DATA_REQ_ByRef.Request.Invite;

          if (p_IsEmergencyCallWithoutRegistration) {      // emergency call without registration   @sic R5s130756 change 14 sic@

            v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_InviteRequest.msgHeader);   /* @sic R5s150268 change 1 sic@ */

            f_IMS_PTC_ImsInfo_Init(v_IMS_DATA_REQ_ByRef.RoutingInfo.UE_Address, v_IMS_DATA_REQ_ByRef.RoutingInfo.NW_Address);

            f_IMS_PortsAndSecurityConfig(f_SIP_ContactUrl_GetPort(v_ContactUrl));          /* @sic R5s150268 change 1 sic@ */

          }

          if (not p_IsReInvite) {      /* @sic R5s140020 change 8.2 sic@ */

            f_IMS_PTC_ImsInfo_DialogInit(dialogRemote, v_InviteRequest.msgHeader);

          } else {

            if (not isvalue(f_IMS_PTC_ImsInfo_CseqGet(dialogRemote))) {  /* @sic R5s140235 sic@ */

              f_IMS_PTC_ImsInfo_CseqSet(dialogRemote, v_InviteRequest.msgHeader.cSeq.seqNumber);

            }

          }

          return v_IMS_DATA_REQ_ByRef;

        }

      [v_TimerControl] t_Timer.timeout {}

    }

    return omit;

  }


Spilt of f_IMS_INVITE_ReceiveRequest_Common
	  function f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(INVITE_A_2_1_Context_Type p_Context,

                                              template (present) SipUrl p_ExpectedCalleeUri,
                                              template (present) MessageHeader p_MessageHeader,



                                              template (omit) float p_WaitDuration := omit,

                                              boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                              boolean p_A8 := false,

                                              boolean p_A11 := false,

                                              boolean p_SrvccAlertingSupported := pc_IMS_SRVCCAlert,

                                              boolean p_OriginatingIdentificationRestriction := false)

    runs on IMS_PTC return template (omit) IMS_DATA_REQ

  { 
    var template (omit) IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (present) MessageBody v_MessageBody := f_SIP_MessageBodySDP_RX(p_A8);  /* @sic R5w140112 sic@ */

    var template (present) MessageHeader v_MessageHeader_Invite;
    var boolean v_IsReInvite := (p_Context == A_2_1_A5);
    var boolean v_IsEmergencyCallWithoutRegistration := (p_Context == A_2_1_A6);
























    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest_Optional(p_ExpectedCalleeUri, p_MessageHeader_Invite, v_MessageBody, p_WaitDuration, v_IsReInvite, v_IsEmergencyCallWithoutRegistration);
    if (isvalue(v_IMS_DATA_REQ)) {
      f_IMS_A_2_1_Invite_CommonChecks(valueof(v_IMS_DATA_REQ), p_Context, p_A3, p_A8, -, p_A11, p_SrvccAlertingSupported, p_OriginatingIdentificationRestriction);
    }


























    return v_IMS_DATA_REQ;
  }


Change 2
f_IMS_InviteRequest_MessageHeaderRX
	  function f_IMS_InviteRequest_MessageHeaderRX(
template (present) SipUrl p_ExpectedCalleeUri,

                                               OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                               boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                               boolean p_A8 := false,

                                               boolean p_A11 := false)

    runs on IMS_PTC return template (present) MessageHeader

  {
    var boolean v_IsGiba := f_IMS_PTC_SecurityScheme_IsGiba();  // => A1/A2

    var boolean v_A17 := f_IMS_PTC_SecurityScheme_IsSipDigest();


    var template (present) MessageHeader v_MessageHeader_Invite := cr_MessageHeader_Dummy;
    var template (present) RouteBody_List v_RouteBodyList;

    var IMS_ProtectedPorts_Type v_Protected;

    var template integer v_Port_us_Contact := *;

    var template integer v_Port_ps_Route := *;

    var template DisplayName v_DisplayName := *;

    var template (present) SipUrl v_SipUrlFrom := ?;    /* per default any SIP URI is accepted in the From header in the first place;

                                                           further checks may be applied in f_IMS_A_2_1_Invite_CommonChecks (depending on the conditions acc. to A.2.1 */

    var template (present) charstring v_Pcscf;

    var charstring v_MediaType := c_sdpApplication;

    var template (present) CallId v_CallId;

    var template (present) integer v_CSeqValue := ?;

    var template (present) From v_From;

    var template (present) To v_To;

    var template AcceptBody_List v_AcceptBodyList := {};

    var OptionTag_List v_OptionTag_List := tsc_OptionTagList_100rel & p_AdditionalOptionTagsForSupported;

    var boolean v_GRUUsInSIP := pc_IMS_GRUUsInSIP;

    var boolean v_A4 := (p_Context == A_2_1_A4);    // @sic R5-166280 sic@

    var boolean v_A5 := (p_Context == A_2_1_A5);    // @sic R5-166280 sic@

    var boolean v_A6 := (p_Context == A_2_1_A6);    // @sic R5-166280 sic@

    var boolean v_A19 := (p_Context == A_2_1_A19);  // @sic R5-174545 sic@

    var boolean v_A26 := pc_IMS_Session_Timer;      // @sic R5-192321 sic@

…

    return v_MessageHeader_Invite;

  }


Change 3
f_IMS_INVITE_ReceiveRequest
	  function f_IMS_INVITE_ReceiveRequest(INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,

                                       OptionTag_List p_AdditionalOptionTagsForSupported := tsc_OptionTagList_Empty,

                                       template SipUrl p_OtherCalleeUri := omit,

                                       boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */

                                       boolean p_A8 := false,

                                       boolean p_A11 := false,

                                       boolean p_SrvccAlertingSupported := pc_IMS_SRVCCAlert)

    runs on IMS_PTC return IMS_DATA_REQ

  { /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    /* @sic R5s140370 change 7: pc_IMS_SRVCCAlert default value for p_A12, p_A12 renamed to p_SrvccAlertingSupported  sic@ */

    /* @sic R5s150180: p_A10 removed sic@ */

    /* @sic R5s150204: p_OtherCalleeUri charstring -> SipUrl sic@ */

    /* @sic R5s150846: p_A8 := false per default sic@ */

    /* @sic R5s170305: p_A9 := pc_IMS_SessionIdSupported per default sic@ */

    /* @sic R5-174544: SessionId removed sic@ */

    var template (omit) float v_WaitDuration := omit;   // no wait duration

    var template (omit) IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (present) SipUrl v_ExpectedCalleeUri := f_IMS_INVITE_CalleeUriRX(p_OtherCalleeUri, p_Context);
    var template (present) MessageHeader v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(v_ExpectedCalleeUri, p_AdditionalOptionTagsForSupported, p_Context, p_A3, p_A8, p_A11);

    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(p_Context, v_ExpectedCalleeUri, v_MessageHeader_Invite, v_WaitDuration, p_A3, p_A8, p_A11, p_SrvccAlertingSupported);
    return valueof(v_IMS_DATA_REQ);                     // as we don't have any WaitDuration f_IMS_INVITE_ReceiveRequest_Common returns a value (omit could only occur when a timer expires)

  }


f_IMS_MOCallSetup_ReceiveINVITE_Optional
	  function f_IMS_MOCallSetup_ReceiveINVITE_Optional(template (omit) float p_WaitDuration := omit,

                                                    template SipUrl p_OtherCalleeUri := omit,

                                                    template SDP_Message p_SDP_Message := omit,

                                                    IMS_SDP_AudioCodecList_Type p_AudioCodecList,

                                                    boolean p_PreconditionSupported := true,

                                                    boolean p_OIR := false,

                                                    boolean p_GeolocationInfoAvailable := false) runs on IMS_PTC return template (omit) IMS_InviteRequestWithSdp_Type

  {
    var OptionTag_List v_AdditionalOptionTagsForSupported := f_IMS_AdditionalOptionTagsForSupported(p_PreconditionSupported);  /* @sic R5s180197 sic@ */

    var template (omit) IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (omit) IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp := omit;

    var template (present) SDP_Message v_SDP_MessageTemplate;

    var IMS_DATA_REQ v_IMS_DATA_REQ_Value;

    var SDP_Message v_SDP_Message;

    var INVITE_A_2_1_Context_Type v_Context := A_2_1_A4;

    var template (present) SipUrl v_ExpectedCalleeUri := f_IMS_INVITE_CalleeUriRX(p_OtherCalleeUri, v_Context);
    var template (present) MessageHeader v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(v_ExpectedCalleeUri, v_AdditionalOptionTagsForSupported, v_Context, -, p_GeolocationInfoAvailable);
    var INVITE_Request v_InviteRequest;

    var boolean v_IsFBBA := f_IMS_PTC_RanTypeIsFBBA();

    var boolean v_CheckForbiddenCodecParameters;                /* @sic R5-180231 sic@ */

    if (ispresent(p_SDP_Message)) {                             /* specific SDP message content @sic R5s180197 sic@ */

      v_SDP_MessageTemplate := p_SDP_Message;

      v_CheckForbiddenCodecParameters := false;                 /* no automatic check for forbidden parameters in case of specific (i.e. non C.21(abcd)) SDP messages */

    }

    else {                                                      /* SDP message content according to C.21(abcd) */

      v_SDP_MessageTemplate := f_IMS_BuildSDP_AnnexC21_Step2(p_PreconditionSupported, p_AudioCodecList); /* @sic R5s150704 - additional changes sic@ @sic R5-172931 sic@ */

      if (v_IsFBBA) {

        v_CheckForbiddenCodecParameters := false;               /* no automatic check for forbidden parameters for FBBA (C.21b, C21c) */

      } else {

        v_CheckForbiddenCodecParameters := true;                /* automatic check for forbidden parameters for Cat-M1 (C.21d), EVS (C.44), WLAN (C.21a) and normal cases (C.21);

                                                                   @sic R5-180231: grace period, R5-192309: grace period for normal cases expired sic@ */

      }

    }

    // Step 2. Receive INVITE Request


    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(v_Context, v_ExpectedCalleeUri, v_MessageHeader_Invite, p_WaitDuration, -, p_GeolocationInfoAvailable, -, -, p_OIR);
    if (isvalue(v_IMS_DATA_REQ)) {

      v_IMS_DATA_REQ_Value := valueof(v_IMS_DATA_REQ);

      v_InviteRequest := v_IMS_DATA_REQ_Value.Request.Invite;

      v_SDP_Message := f_IMS_MessageBody_CheckSdpMessageAndGeolocation(v_InviteRequest.messageBody, v_SDP_MessageTemplate, p_GeolocationInfoAvailable);

      // check media description:

      f_IMS_SDP_CheckMediaDescr_AudioCodecs(v_SDP_Message.media_list[0], p_AudioCodecList, v_CheckForbiddenCodecParameters);  /* @sic R5-151968: f_IMS_SDP_CheckMediaDescr_EVS_16000_1, f_IMS_SDP_CheckMediaDescr_AMR_16000_1 replaced by f_IMS_SDP_CheckMediaDescr_AudioCodec sic@

                                                                                                                                 @sic R5-172931: f_IMS_SDP_CheckMediaDescr_AudioCodecs instead of f_IMS_SDP_CheckMediaDescr_AudioCodec sic@

                                                                                                                                 @sic R5-180231: v_CheckForbiddenCodecParameters sic@ */

      v_InviteRequestWithSdp := f_IMS_InviteRequestWithSdpMO(v_IMS_DATA_REQ_Value, v_SDP_Message);

    }

    return v_InviteRequestWithSdp;

  }


f_IMS_SRVCC_ReInvite_AnnexC24_Step1
	  function f_IMS_SRVCC_ReInvite_AnnexC24_Step1(charstring p_OtherCalleeUri,

                                               template (omit) float p_WaitDuration := omit,

                                               boolean p_GeolocationInfoAvailable := false)

    runs on IMS_PTC return template (omit) IMS_InviteRequestWithSdp_Type

  { /* @sic R5w140112: return type changed to IMS_InviteRequestWithSdp_Type to carry decoded SDP offer sic@ */

    /* @sic R5s140331 change 8: parameter p_WaitDuration sic@ */

    /* @sic R5-151957: parameter p_OtherCalleeUri sic@ */

    var template (present) SDP_Message v_SDP_MessageRX := f_IMS_BuildSDP_AnnexC24_Step1();

    var template (omit) IMS_DATA_REQ v_IMS_DATA_REQ;

    var IMS_DATA_REQ v_IMS_DATA_REQ_Val;

    var SDP_Message v_SDP_Message;

    var INVITE_A_2_1_Context_Type v_Context := A_2_1_A5;
    var template (present) SipUrl v_CalleeSipUrl := f_SIP_BuildSipUri_RX(p_OtherCalleeUri, *);  /* @sic R5-151957 sic@ */

    var template (present) MessageHeader v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(v_CalleeSipUrl, -, v_Context, -, p_GeolocationInfoAvailable);

    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(v_Context, v_CalleeSipUrl, v_MessageHeader_Invite, p_WaitDuration, -, p_GeolocationInfoAvailable);
    if (isvalue(v_IMS_DATA_REQ)) {

      v_IMS_DATA_REQ_Val := valueof(v_IMS_DATA_REQ);

      v_SDP_Message := f_IMS_MessageBody_CheckSdpMessageAndGeolocation(v_IMS_DATA_REQ_Val.Request.Invite.messageBody, v_SDP_MessageRX, p_GeolocationInfoAvailable);

      return f_IMS_InviteRequestWithSdpMO(v_IMS_DATA_REQ_Val, v_SDP_Message);

    }

    return omit;

  }


f_IMS_SRVCC_ReInvite_AnnexC31_Step1
	  function f_IMS_SRVCC_ReInvite_AnnexC31_Step1(charstring p_OtherCalleeUri,

                                               boolean p_GeolocationInfoAvailable := false,

                                               template (omit) float p_WaitDuration := omit)  // @sic R5-174016 sic@

    runs on IMS_PTC return template (omit) IMS_InviteRequestWithSdp_Type

  { /* @sic R5s140020, R5-140799: new function sic@ */

    /* @sic R5-140799 sic@ */

    /* @sic R5w140112: return type changed to IMS_InviteRequestWithSdp_Type to carry decoded SDP offer */

    /* @sic R5-150706: new parameter p_OtherCalleeUri to hand over the contact URI being used in the intial INVITE (MT call) sic@ */

    var template (omit) float v_WaitDuration := p_WaitDuration;   //  @sic R5-174016 sic@

    var template (present) SDP_Message v_SDP_MessageRX := f_IMS_BuildSDP_AnnexC31_Step1();


    var template (omit) IMS_DATA_REQ v_IMS_DATA_REQ;

    var IMS_DATA_REQ v_IMS_DATA_REQ_Val;

    var SDP_Message v_SDP_Message;

    var INVITE_A_2_1_Context_Type v_Context := A_2_1_A5;
    var template (present) SipUrl v_CalleeSipUrl := f_SIP_BuildSipUri_RX(p_OtherCalleeUri, *);    /* @sic R5s150204 sic@ */

    var template (present) MessageHeader v_MessageHeader_Invite;

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(v_CalleeSipUrl, -, v_Context, -, p_GeolocationInfoAvailable);

    v_MessageHeader_Invite.reason := f_IMS_SIP_Reason487_RX();     // @sic R5s140370 change 3 - MCC160 implementation sic@


    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(v_Context, v_CalleeSipUrl, v_MessageHeader_Invite, v_WaitDuration, -, p_GeolocationInfoAvailable);
    if (isvalue(v_IMS_DATA_REQ)) { // @sic R5-174016 sic@

      v_IMS_DATA_REQ_Val := valueof(v_IMS_DATA_REQ);

      v_SDP_Message := f_IMS_MessageBody_CheckSdpMessageAndGeolocation(v_IMS_DATA_REQ_Val.Request.Invite.messageBody, v_SDP_MessageRX, p_GeolocationInfoAvailable);

      return f_IMS_InviteRequestWithSdpMO(v_IMS_DATA_REQ_Val, v_SDP_Message);

    }

    return omit;

  }


Change 4
f_IMS_InviteResponse_183_MessageHeaderTX
	  function f_IMS_InviteResponse_183_MessageHeaderTX(INVITE_Request p_InviteRequest,

                                                    OptionTag_List p_OptionTagsForRequire := tsc_OptionTagList_100rel_precondition,

                                                    charstring p_ContactUri,

                                                    template (omit) SemicolonParam_List p_AditionalContactParams := omit,

                                                    IMS_CallType_Type p_CallType := NORMAL_CALL,

                                                    boolean p_A6 := pc_IMS_Video_FeatureTag
,

                                                    template (omit) ContentType p_ContentType := cs_ContentTypeSDP) runs on IMS_PTC return template (value) MessageHeader

  { 

    var MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (value) MessageHeader v_MessageHeader_Response := cs_MessageHeader_Dummy;

    var template (omit) SemicolonParam_List v_ContactParams := p_AditionalContactParams;


    var template (value) RouteBody_List v_RouteBodyList := f_IMS_RouteSet_MO_Call_TX(p_CallType);     /* @sic R5s150692 change 7 sic@ */

    v_MessageHeader_Response.recordRoute     := cs_RecordRoute(v_RouteBodyList);

    v_MessageHeader_Response.via             := v_MessageHeader_Invite.via;

    v_MessageHeader_Response.require         := cs_Require(v_OptionTag_List); // @sic R5s130333 change 3 sic@

    v_MessageHeader_Response.fromField       := v_MessageHeader_Invite.fromField;

    v_MessageHeader_Response.toField         := f_IMS_Response_ToHeaderTX(v_MessageHeader_Invite); /* @sic R5-131897, R5-132063 - generic handling of to-tags sic@ */


    v_MessageHeader_Response.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Response.cSeq            := v_MessageHeader_Invite.cSeq;

    v_MessageHeader_Response.contentType     := p_ContentType;
    if (lengthof(p_OptionTagsForRequire) > 0) {
      v_MessageHeader_Response.require       := cs_Require(p_OptionTagsForRequire);
      if (f_MessageHeader_CheckRequire(valueof(v_MessageHeader_Response), c_tag100rel)) {
        v_MessageHeader_Response.rSeq        := cs_RSeq(tsc_IMS_RSeqNumFor183);

      }

    }
    if (p_CallType == ASSERTED_EMERGENCY) {    // Condition A5 @sic R5s150692 change 7 sic@

      v_MessageHeader_Response.pAssertedID   := cs_PAssertedId(f_SIP_BuildSipUri_lr_TX(tsc_IMS_Emergency_TelUri));

    }

    if (pc_BSRVCC) {  // @sic R5s140962 sic@

      v_MessageHeader_Response.featureCaps   := cs_FeatureCaps({ cs_GenericParam(tsc_IMS_FeatureCap_SrvccPreAlerting)});  // @sic R5-145796 sic@

    }

    if (p_A6) { // @sic R5s130109 sic@

      v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("video"));

    }

    v_ContactParams := f_SemicolonParam_List_Add_TX(v_ContactParams, cs_GenericParam("audio"));                           // @sic R5-165943 sic@

    v_MessageHeader_Response.contact         := cs_Contact(f_SIP_BuildSipUri_TX(p_ContactUri), v_ContactParams);




    return v_MessageHeader_Response;

  }


tsc_OptionTagList_100rel_precondition
	  const OptionTag_List tsc_OptionTagList_100rel_precondition := tsc_OptionTagList_100rel & tsc_OptionTagList_precondition;


fl_IMS_MOCallSetup_CAT_Step9
	  function fl_IMS_MOCallSetup_CAT_Step9(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,

                                        charstring p_ContactUri) runs on IMS_PTC return MessageHeader

  {

    var boolean v_Earlymedia := true;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_InviteRequestWithSdp.RoutingInfo);

    var INVITE_Request v_InviteRequest := p_InviteRequestWithSdp.Invite;

    var SDP_Message v_SDP_MessageInvite := p_InviteRequestWithSdp.SdpOffer;

    var template (value) MessageHeader v_MessageHeader_Response183;

    var template (value) SDP_Message v_SDP_MessageTx := fl_IMS_BuildSDP_CAT_Step9(v_SDP_MessageInvite);

    var template (value) SemicolonParam_List v_AditionalContactParams := { cs_GenericParam_Common(tsc_IMS_FeatureParamId_3gpp_icsi_ref, cs_GenValueQuoted(cs_IMS_EncodedFeatureParamValue_3gpp_icsi_ref)) };

    // Step 9. Send 183 Session In Progress.

    v_MessageHeader_Response183 := f_IMS_InviteResponse_183_MessageHeaderTX(v_InviteRequest, -, p_ContactUri, v_AditionalContactParams, -, -, v_Earlymedia); //CAT specific

    v_MessageHeader_Response183.pEarlyMedia  := cs_PEarlyMedia({c_sendonly});
    v_MessageHeader_Response183.toField.toParams[0].paramValue.tokenOrHost := "xyzTC20_1-INVITE-Totag"; /* @sic R5s150714: dialog 2 setting see 4.2  sic@ */

    f_IMS_PTC_ImsInfo_DialogSetLocalTag("xyzTC20_1-INVITE-Totag");

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine183, v_MessageHeader_Response183, cs_MessageBody_SDP(v_SDP_MessageTx))));

    return valueof(v_MessageHeader_Response183);

  }


Change 5
f_IMS_InviteRequest_MessageHeaderTX_Common
	  function f_IMS_InviteRequest_MessageHeaderTX_Common(OptionTag_List p_OptionTagsForSupported := tsc_OptionTagList_Invite_C11,

                                                      boolean p_IsReInvite := false, 
                                                      charstring p_CallerSipAddrStr := tsc_IMS_CallerSipAddrStr,
                                                      integer p_CSeqValue := 4711) runs on IMS_PTC return template (value) MessageHeader

  {

    var template (value) MessageHeader v_MessageHeader_Invite := cs_MessageHeader_Dummy;

    var template (value) SipUrl v_PublicUserId_SipUrl := f_IMS_PTC_ImsInfo_GetSipUriTX(defaultId);  // @sic R5-133304: use default public user id sic@

    var charstring v_CallIdString := f_IMS_PTC_GenerateCallId("INVITE");

    var charstring v_FromTag := f_IMS_GenerateTag("INVITE", "From");

    var template (value) RouteBody_List v_RouteBodyList;

    var integer v_CSeqValue;

    var template (value) From v_From;

    var template (value) To v_To;

    var template (value) CallId v_CallId; 
    var SipUrl v_CallerSipUri;

    var charstring v_CallerDomainName;
    if (not p_IsReInvite) {

      v_CSeqValue := p_CSeqValue;

      v_From   := cs_From(f_SIP_BuildSipUri_TX(p_CallerSipAddrStr), v_FromTag);

      v_To     := cs_ToDef(v_PublicUserId_SipUrl);

      v_CallId := cs_CallId(v_CallIdString);

      v_RouteBodyList := f_IMS_RouteSet_MT_Call_TX();      /* @sic R5s140324 - MCC160 Implementation sic@ */

    }

    else {

      v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogLocal);

      v_From   := f_IMS_RequestInDialog_FromHeaderTX();

      v_To     := f_IMS_RequestInDialog_ToHeaderTX();

      v_CallId := f_IMS_PTC_ImsInfo_DialogGetCallId();

      if (f_IMS_PTC_ImsInfo_DialogIsMO()) {      /* @sic R5s140324 - MCC160 Implementation sic@ */

        v_RouteBodyList := fl_IMS_RouteSet_ReverseOrder(f_IMS_RouteSet_MO_Call_TX());    /* @sic R5-142320, R5-144601 sic@ */

      } else {

        v_RouteBodyList := f_IMS_RouteSet_MT_Call_TX();

      }

    } 
    v_CallerSipUri := f_Addr_Union_GetSipUrl(v_From.addressField);

    v_CallerDomainName := v_CallerSipUri.components.sip.hostPort.host;
    v_MessageHeader_Invite.via             := f_IMS_InviteRequest_ViaTX(v_CallerDomainName);  /* @sic R5-134116: New function fl_IMS_InviteRequest_ViaTX sic@ */

    v_MessageHeader_Invite.recordRoute     := cs_RecordRoute(v_RouteBodyList);

    v_MessageHeader_Invite.fromField       := v_From;

    v_MessageHeader_Invite.toField         := v_To;

    v_MessageHeader_Invite.callId          := v_CallId;

    v_MessageHeader_Invite.cSeq            := cs_CseqDef(v_CSeqValue, "INVITE");

    v_MessageHeader_Invite.supported       := cs_Supported(p_OptionTagsForSupported);

    v_MessageHeader_Invite.pCalledPartyID  := cs_PCalledPartyId(v_PublicUserId_SipUrl);

    v_MessageHeader_Invite.maxForwards     := cs_MaxForwardsDef;

    if (not p_IsReInvite) {

      f_IMS_PTC_ImsInfo_DialogInit(dialogLocal, valueof(v_MessageHeader_Invite));

    }

    f_IMS_PTC_ImsInfo_DialogSetRemoteRSeq(omit);       /* reset RSeq for every (re-)INVITE   @sic R5-181190 sic@ */

    return v_MessageHeader_Invite;

  }


f_IMS_ByeRequest_MessageHeaderTX
	  function f_IMS_ByeRequest_MessageHeaderTX(template (value) INVITE_Request p_InviteRequest,

                                            template (omit) ContentType p_ContentType := omit) runs on IMS_PTC return template (value) MessageHeader

  { 
    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (value) MessageHeader v_MessageHeader_Bye := cs_MessageHeader_Dummy;

    var template (value) Via v_Via; 
    var SipUrl v_CallerSipUri;

    var charstring v_CallerDomainName; 
    var template (value) From v_From := f_IMS_RequestInDialog_FromHeaderTX();
    var integer v_CSeqValue := f_IMS_PTC_ImsInfo_CseqIncr(dialogLocal);

    var boolean v_IsMCX := f_IMS_PTC_TestsuiteInfo_IsMCX();

    if (fl_IMS_InviteRequestIsMO(p_InviteRequest)) {                    /* @sic R5-134116: distinguish MO/MT call establishment sic@ */ 
      v_CallerSipUri := f_Addr_Union_GetSipUrl(v_From.addressField);

      v_CallerDomainName := v_CallerSipUri.components.sip.hostPort.host;
      v_Via := f_IMS_InviteRequest_ViaTX(v_CallerDomainName);

    } else {

      v_Via := f_IMS_Via_AddSigCompParam(v_MessageHeader_Invite.via);

      v_Via := f_IMS_ViaRefreshBranches(v_Via);                         /* @sic R5s130510 sic@ */

    }

    v_MessageHeader_Bye.via             := v_Via;

    v_MessageHeader_Bye.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Bye.cSeq            := cs_CseqDef(v_CSeqValue, "BYE");

    v_MessageHeader_Bye.maxForwards     := cs_MaxForwardsDef;

    v_MessageHeader_Bye.toField         := f_IMS_RequestInDialog_ToHeaderTX();
    v_MessageHeader_Bye.fromField       := v_From;

    v_MessageHeader_Bye.contentType     := p_ContentType;

    if (v_IsMCX) {                                                      /* @sic R5-196983 sic@ */

      v_MessageHeader_Bye.pAssertedID   := cs_PAssertedId(f_SIP_BuildSipUri_TX(f_IMS_PTC_TestsuiteInfo_GetParameter("MCPTT_Server_A_URI")));   // !!!! PROSE ISSUE: comment says "The URI of the SS" !!!!

    }

    return v_MessageHeader_Bye;

  }


Change 6
cr_FeatureParam_3gpp_icsi_ref
	

  template (value) GenericParam crs_FeatureParam_3gpp_icsi_ref(template (value) charstring p_QuotedString := crs_IMS_EncodedFeatureParamValue_3gpp_icsi_ref) := 
    cs_GenericParam_Common(tsc_IMS_FeatureParamId_3gpp_icsi_ref, cs_GenValueQuoted(p_QuotedString));


Change 7
f_IMS_InviteResponse_183_MessageHeaderRX
	  function f_IMS_InviteResponse_183_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,


                                                    template ContentType p_ContentType := cr_ContentTypeSDP,
                                                    OptionTag_List p_OptionTagsForRequire := tsc_OptionTagList_100rel_precondition,
                                                    template SemicolonParam_List p_MandatoryContactParams := omit,
                                                    boolean p_A6 := pc_IMS_Video_FeatureTag)

    runs on IMS_PTC return template (present) MessageHeader

  { 
    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Response := cr_MessageHeader_Dummy;

    var integer v_Port_us := f_IMS_PTC_GetPort_us();


    var template (present) SemicolonParam_List v_MandatoryContactParams := { };

    if (ispresent(p_MandatoryContactParams)) {                          /* @sic R5-196983: support of MCX sic@ */

      v_MandatoryContactParams := p_MandatoryContactParams;             /* causes compiler warning */

    } 

    else {
      v_MandatoryContactParams := { crs_FeatureParam_3gpp_icsi_ref };
      if (p_A6) {  /* @sic R5s150180 sic@ */

        v_MandatoryContactParams := f_SemicolonParam_List_Add_RX(v_MandatoryContactParams, cr_FeatureParam_Video);

      }

      if (f_IMS_PTC_AudioFeatureTagIsRequired() ) {                                                                    // @sic R5-173021 sic@ @sic R5-177127 sic@

        v_MandatoryContactParams := f_SemicolonParam_List_Add_RX(v_MandatoryContactParams, cr_FeatureParam_Audio);     // @sic R5-165943, R5-170564 sic@

      }
    }
    v_MessageHeader_Response.via          := f_Via_ResponseRX(v_MessageHeader_Invite.via);                           /* @sic R5s140350 sic@ */

    v_MessageHeader_Response.recordRoute  := f_IMS_RecordRoute_AsInInvite_RX(v_MessageHeader_Invite.recordRoute);    /* @sic R5s130510 change 6.1, R5-135004 change 1 sic@

                                                                                                                        @sic R5s150039 change 2 sic@ */

    v_MessageHeader_Response.require      := cr_Require(p_OptionTagsForRequire);

    v_MessageHeader_Response.fromField    := f_IMS_Response_FromHeaderRX(v_MessageHeader_Invite);  /* @sic R5s130453  sic@ */

    v_MessageHeader_Response.toField      := f_IMS_Response_ToHeaderRX(v_MessageHeader_Invite);    /* @sic R5s130453  sic@ */

    v_MessageHeader_Response.contact      := cr_Contact(f_Contact_SipUri_HostPortRX(?, v_Port_us), v_MandatoryContactParams);    /* host address needs to be checked after receiving (-> a_IMS_ReceiveResponse)

                                                                                                                                    @sic R5s150721, R5-153760: cr_SipUri_HostPort replaced by f_Contact_SipUri_HostPortRX sic@ */


    v_MessageHeader_Response.callId       := v_MessageHeader_Invite.callId;

    v_MessageHeader_Response.cSeq         := v_MessageHeader_Invite.cSeq;


    v_MessageHeader_Response.contentType  := p_ContentType;
    return v_MessageHeader_Response;

  }


a_IMS_ReceiveResponse
	  altstep a_IMS_ReceiveResponse(template (value) MessageHeader p_RequestHeader,

                                template (present) InternetProtocol_Type p_Protocol,

                                template (present) IMS_Response_Type p_Response,

                                out IMS_DATA_RSP p_IMS_DATA_RSP_ByRef) runs on IMS_PTC

  {

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    var MessageHeader v_MessageHeader;

    var template (omit) integer v_PrevRSeq;

    [] IMS_Client.receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(p_Protocol), p_Response)) -> value v_IMS_DATA_RSP

      {

        if ((v_IMS_DATA_RSP.RoutingInfo.Protocol == tcp) and ispresent(v_IMS_DATA_RSP.Response.messageBody)) {     /* @sic R5-169122 sic@ */

          if (not match(v_IMS_DATA_RSP.Response.msgHeader, cr_MessageHeader_Dummy(cr_ContentLength()))) {

            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Response sent on tcp shall have a ContentLength when having a MessageBody");

          }

        }

        v_MessageHeader := v_IMS_DATA_RSP.Response.msgHeader;

        f_IMS_MessageHeader_Response_CheckVia(valueof(p_RequestHeader), v_MessageHeader);

        select (v_IMS_DATA_RSP.Response.statusLine.statusCode) {                                /* @sic R5-181190: store and check RSeq value for reliable responses sic@ */

          case (180, 183) {

            fl_IMS_MessageHeader_180_183_Response_CommonChecks(v_MessageHeader);                /* @sic R5s180202 sic@ */





            if (f_MessageHeader_CheckRequire(v_MessageHeader, c_tag100rel)) {

              if (not ispresent(v_MessageHeader.rSeq.responseNum)) {

                f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "reliable 180 response shall have RSeq");                /* 34.229-1 A.2.6 condition A3, A12 */

              }

              v_PrevRSeq := f_IMS_PTC_ImsInfo_DialogGetRemoteRSeq();

              if (isvalue(v_PrevRSeq) and (v_MessageHeader.rSeq.responseNum != (valueof(v_PrevRSeq) + 1))) {

                f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "RSeq value needs to be previous RSeq value + 1");       /* 34.229-1 A.2.6 condition A3 */

              }

              f_IMS_PTC_ImsInfo_DialogSetRemoteRSeq(v_MessageHeader.rSeq.responseNum);

            }

          }

        }

        p_IMS_DATA_RSP_ByRef := v_IMS_DATA_RSP;

      }

  }


Change 8
f_IMS_InviteResponse_180_MessageHeaderRX
	  function f_IMS_InviteResponse_180_MessageHeaderRX(template (value) INVITE_Request p_InviteRequest,

 
                                                    template ContentType p_ContentType := *,
                                                    OptionTag_List p_OptionTagsForRequire := tsc_OptionTagList_Empty, 
                                                    template SemicolonParam_List p_MandatoryContactParams := omit,
                                                    boolean p_SrvccAlerting := pc_IMS_SRVCCAlert) runs on IMS_PTC return template (present) MessageHeader

  {
    var template (value) MessageHeader v_MessageHeader_Invite := p_InviteRequest.msgHeader;

    var template (present) MessageHeader v_MessageHeader_Response := cr_MessageHeader_Dummy;

    var integer v_Port_us := f_IMS_PTC_GetPort_us();

    var template (present) SemicolonParam_List v_MandatoryContactParams := {};

    var OptionTag_List v_MandatoryOptions;

    var boolean v_IsMCX := f_IMS_PTC_TestsuiteInfo_IsMCX();

   if (ispresent(p_MandatoryContactParams)) {                          /* @sic R5-196983: support of MCX sic@ */

      v_MandatoryContactParams := p_MandatoryContactParams;             /* causes compiler warning */

    }
    else {

      if (f_IMS_PTC_AudioFeatureTagIsRequired() ) {                                                                    // @sic R5-173021 sic@ @sic R5-177127 sic@

        v_MandatoryContactParams := f_SemicolonParam_List_Add_RX(v_MandatoryContactParams, cr_FeatureParam_Audio);     // @sic R5-165943, R5-170564 sic@

      }

      if (p_SrvccAlerting) {

        v_MandatoryContactParams := f_SemicolonParam_List_Add_RX(v_MandatoryContactParams, cr_GenericParam_NoValue(tsc_IMS_FeatureCap_SrvccAlerting));

      }

    } 

    if (lengthof(p_OptionTagsForRequire) > 0) { 
      v_MessageHeader_Response.require       := cr_Require(p_OptionTagsForRequire);

    }
    v_MessageHeader_Response.recordRoute     := f_IMS_RecordRoute_AsInInvite_RX(v_MessageHeader_Invite.recordRoute);     /* @sic R5s130510 change 9 sic@ @sic R5s150039 change 2 sic@ */

    v_MessageHeader_Response.via             := f_Via_ResponseRX(v_MessageHeader_Invite.via);         /* @sic R5s140350 sic@ */

    v_MessageHeader_Response.fromField       := f_IMS_Response_FromHeaderRX(v_MessageHeader_Invite);  /* @sic R5s130453 sic@ */

    v_MessageHeader_Response.toField         := f_IMS_Response_ToHeaderRX(v_MessageHeader_Invite);    /* @sic R5s130453 sic@ */

    v_MessageHeader_Response.contact         := cr_Contact(f_Contact_SipUri_HostPortRX(?, v_Port_us), v_MandatoryContactParams);  /* host address needs to be checked after receiving (-> a_IMS_ReceiveResponse)

                                                                                                                                     @sic R5s150721, R5-153760: cr_SipUri_HostPort replaced by f_Contact_SipUri_HostPortRX sic@

                                                                                                                                     @sic R5s170216: v_MandatoryContactParams sic@ */

    v_MessageHeader_Response.callId          := v_MessageHeader_Invite.callId;

    v_MessageHeader_Response.cSeq            := v_MessageHeader_Invite.cSeq;

    v_MessageHeader_Response.contentType     := p_ContentType;                      /* @sic R5-140975 sic@ */

    /* @sic R5-169122: contentLength removed sic@ */

    /* for 180 Ringing being sent reliable rSeq is checked in a_IMS_ReceiveResponse */

    v_MessageHeader_Response.pAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());    // NOTE: acc. A.2.6 there is no condition regarding GIBA




    return v_MessageHeader_Response;

  }


Change 9
f_IMS_RegisterResponse_Base_MessageHeaderTX
	  function f_IMS_BaseResponse_MessageHeaderTX (REGISTER_Request p_RegisterRequest ,

                                              template (omit) charstring p_ToTag := omit) runs on IMS_PTC return template (value) MessageHeader

  { 

    var MessageHeader v_MessageHeader_Register := p_RegisterRequest.msgHeader;

    var template (value) MessageHeader v_MessageHeader_Response := cs_MessageHeader_Dummy;

    v_MessageHeader_Response.via             := v_MessageHeader_Register.via;

    v_MessageHeader_Response.toField         := f_IMS_Response_ToHeaderTX(v_MessageHeader_Register, p_ToTag);
    v_MessageHeader_Response.fromField       := v_MessageHeader_Register.fromField;

    v_MessageHeader_Response.callId          := v_MessageHeader_Register.callId;

    v_MessageHeader_Response.cSeq            := v_MessageHeader_Register.cSeq;

    return v_MessageHeader_Response;

  }


Further changes:

f_IMS_BaseResponse_MessageHeaderTX used by 

· f_IMS_InviteResponse_181_MessageHeaderTX

· f_IMS_Accepted_202_MessageHeaderTX

· f_IMS_InviteResponse_380_MessageHeaderTX

· f_IMS_InviteResponse_180_MessageHeaderTX

· f_IMS_InviteResponse_183_MessageHeaderTX

· f_IMS_OtherResponse_200_MessageHeaderTX

· f_IMS_InviteResponse_4XX_MessageHeaderTX

· f_IMS_CommonResponse_503_MessageHeaderTX

· fl_ECall_InviteErrResponse_MessageHeaderTX

· f_IMS_RequestTerminated_487MessageHeaderTX

· fl_MCX_RegisterResponse_403_MessageHeaderTX

· f_IMS_SubscribeOrPublishResponse_200_MessageHeaderTX
